
01 Auf einen Blick
Das ESP32 DevKit bringt alles mit, was ein UNO nicht hat: zwei Kerne mit bis zu 240 MHz,

WLAN und Bluetooth auf demselben Chip. Programmiert wird es wie ein Arduino – mit der

Arduino IDE und einem USB-C-Kabel.

Technische Daten

Modul ESP32-WROOM-32 (Chip

ESP32-D0WD-V3)

Prozessor Dual-Core Xtensa LX6, bis

240 MHz

Speicher 4 MB Flash, 520 kB SRAM

Funk WLAN 2,4 GHz · Bluetooth

Classic + BLE

USB USB-C, Wandler CH340C

Versorgung 5 V über USB-C oder VIN

Logikpegel 3,3 V – Pins nicht 5-V-

tolerant

Anschlüsse 30 Pins (2 × 15), 2,54-mm-

Raster

Lieferumfang

1 × ESP32 CH340C Entwicklungsboard (30 Pins,

USB-C)

! 3,3 V an den Pins

Die GPIOs arbeiten mit 3,3 V und

vertragen keine 5 V. 5-V-Signale nur über

Spannungsteiler oder Pegelwandler

anschließen.

i Strom: USB-C oder VIN

Versorge das Board über USB-C oder mit

5 V an VIN. Der 3V3-Pin ist ein Ausgang

des Spannungsreglers – dort nichts

einspeisen.

Das ist auf der Platine

Taste EN Neustart (Reset) – der Sketch beginnt von vorn.

Taste BOOT Nur nötig, falls ein Upload hängen bleibt (Seite 5).

LED „PWR“ Leuchtet, sobald das Board Strom hat.

LED „IO2“ Blaue LED an GPIO2 – im ersten Sketch blinkt sie.

Chip CH340C USB-Seriell-Wandler: macht aus USB-C einen COM-Port.

Was du zusätzlich brauchst

USB-C-Datenkabel

Kein reines Ladekabel – es muss Daten

übertragen

 

PC mit Arduino IDE 2

Windows, macOS oder Linux, kostenlos

 

Für die Beispiele (optional)

Steckbrett, Jumper, LED + 220 Ω, Poti

10 kΩ, I2C-Sensor

MKR-SET-BB400-JUMPER

MAKEROO · support@makeroo.de ESP32 DevKit · MKR-ESP32-005

02 Pinbelegung
So, wie das Board vor dir liegt: Modul oben, USB-C unten. Die Namen entsprechen dem

Aufdruck auf der Platine, im Code verwendest du die GPIO-Nummer.

Versorgung Masse frei nutzbar ADC1 / DAC nur Eingang I2C SPI USB-Seriell

Strapping – Vorsicht

! Strapping-Pins: D2, D5, D12, D15

Ihr Pegel entscheidet beim Einschalten, wie der ESP32 startet. Schließe dort nichts an, was die

Leitung beim Booten fest auf HIGH oder LOW zieht – sonst schlägt der Upload fehl. Bei aktivem

WLAN funktionieren analoge Messungen nur an ADC1 (D32–D39).

ESP-WROOM-32
Wi-Fi + BT

ESPRESSIF

CH340C

AMS1117

EN BOOT

ESP32 DEVKIT

EN

VP

VN

D34

D35

D32

D33

D25

D26

D27

D14

D12

D13

GND

VIN

D23

D22

TX0

RX0

D21

D19

D18

D5

D17

D16

D4

D2

D15

GND

3V3

Reset (aktiv LOW) EN

nur Eingang · ADC1 GPIO36 VP

nur Eingang · ADC1 GPIO39 VN

nur Eingang · ADC1 GPIO34 D34

nur Eingang · ADC1 GPIO35 D35

ADC1 · Touch GPIO32 D32

ADC1 · Touch GPIO33 D33

DAC1 (Analogausgang) GPIO25 D25

DAC2 (Analogausgang) GPIO26 D26

frei · Touch GPIO27 D27

PWM beim Booten GPIO14 D14

Strapping · Boot: LOW GPIO12 D12

frei GPIO13 D13

Masse GND

5-V-Eingang VIN

D23 GPIO23 SPI MOSI · frei

D22 GPIO22 I2C SCL

TX0 GPIO1 USB-Seriell – belegt

RX0 GPIO3 USB-Seriell – belegt

D21 GPIO21 I2C SDA

D19 GPIO19 SPI MISO

D18 GPIO18 SPI SCK

D5 GPIO5 SPI CS · Strapping

D17 GPIO17 frei · UART2 TX

D16 GPIO16 frei · UART2 RX

D4 GPIO4 frei · ADC2 · Touch

D2 GPIO2 blaue LED · Strapping

D15 GPIO15 Strapping

GND Masse

3V3 3,3-V-Ausgang

MAKEROO · support@makeroo.de ESP32 DevKit · MKR-ESP32-005

03 Einrichten
Diese Schritte machst du nur ein einziges Mal. Die Bilder stammen aus der Arduino IDE 2.3

mit angeschlossenem Board.

1 Arduino IDE installieren

Lade die aktuelle Arduino IDE 2 von arduino.cc/en/software herunter und installiere sie.

2 Board anschließen, Treiber prüfen

Board per USB-C anstecken – die rote LED „PWR“ leuchtet. Windows 10/11 und macOS bringen den

Treiber für den CH340C meist mit. Taucht später kein Port auf, installiere „CH341SER“ vom Chiphersteller

WCH (wch-ic.com).

3 ESP32-Board-Paket installieren

Links auf das Board-Symbol klicken (Boardverwalter), esp32 eingeben und beim Eintrag esp32 von

Espressif Systems auf Installieren klicken. Der Download ist rund 300 MB groß und dauert ein paar

Minuten. Nicht verwechseln mit „Arduino ESP32 Boards“ – das ist für den Arduino Nano ESP32.

4 Board und Port wählen

In der Board-Auswahl oben auf Anderes Board und Port wählen gehen, ESP32 Dev Module suchen und

den neuen COM-Port wählen (Windows: COMx, macOS: /dev/cu.wchusbserial…, Linux: /dev/ttyUSB0).

Unten rechts steht danach z. B. „ESP32 Dev Module an COM7“.

ARDUINO IDE · BOARDVERWALTER

MAKEROO · support@makeroo.de ESP32 DevKit · MKR-ESP32-005

04 Hochladen & Testen
Jetzt kommt dein erster Sketch auf den ESP32: Er lässt die blaue LED blinken und zeigt dir

alle WLAN-Netze in der Umgebung.

5 Sketch hochladen

Neuen Sketch anlegen, den Code von Seite 6 einfügen und mit dem Pfeil-Symbol hochladen ( Strg  +

U ). Das erste Kompilieren dauert etwa eine Minute. Unten erscheinen „Connecting…“, die

Prozentanzeige und am Ende „Hard resetting via RTS pin…“.

6 Seriellen Monitor öffnen

Strg  + Umschalt  + M , unten rechts 115200 Baud einstellen. Alle zehn Sekunden erscheint eine neue

Runde mit Chip-Infos und den gefundenen Netzen. Die blaue LED „IO2“ blinkt vor jedem Scan dreimal.

i Upload hängt bei „Connecting…“?

Taste BOOT gedrückt halten, bis die Prozentanzeige startet, dann loslassen. Bei unserem Board war

das im Test nicht nötig – der CH340C schaltet automatisch in den Upload-Modus.

UPLOAD ERFOLGREICH

SERIELLER MONITOR · 115200 BAUD

MAKEROO · support@makeroo.de ESP32 DevKit · MKR-ESP32-005

05 Der erste Sketch
Kein Zubehör nötig: Der Sketch nutzt nur die blaue LED und das WLAN auf dem Board.

Zeile 4

WLAN-Bibliothek

Gehört zum ESP32-Board-Paket –

nichts extra installieren.

Zeile 6–7

Pins festlegen

GPIO2 ist die blaue LED „IO2“ auf

dem Board, GPIO23 die optionale

LED.

Zeile 14

WLAN einschalten

Als Client, ohne Verbindung – genug,

um Netze zu suchen.

Zeile 19–24

Dreimal blinken

Zeigt sichtbar an, dass dein Sketch

läuft.

Zeile 26–28

Chip-Infos

Modell, Kerne, Takt und Flash liest

der ESP32 aus sich selbst aus.

Zeile 29–34

WLAN-Scan

Listet bis zu sechs Netze mit

Signalstärke (dBm) und Kanal.

Auf diesem Board getestet, Board-Paket esp32 3.3.12, „ESP32 Dev Module“: 860 kB Flash (67 %), 45 kB RAM (13 %).

MAKEROO_ESP32_Start.ino

// MAKEROO · ESP32 DevKit (WROOM-32, CH340C) · Erster Start1

// Blinkt die blaue LED (IO2) und optional eine LED an GPIO23,2

// zeigt Chip-Infos und sucht alle 10 s nach WLAN-Netzen.3
#include <WiFi.h>4

 5

const int LED_BOARD  = 2;   // blaue Onboard-LED "IO2"6

const int LED_EXTERN = 23;  // optionale LED auf dem Steckbrett7

int runde = 0;8
 9

void setup() {10

  Serial.begin(115200);11

  pinMode(LED_BOARD, OUTPUT);12

  pinMode(LED_EXTERN, OUTPUT);13
  WiFi.mode(WIFI_STA);        // WLAN als Client, ohne Verbindung14

  Serial.println("\nMAKEROO ESP32 - Erster Start");15

}16

 17
void loop() {18

  for (int i = 0; i < 6; i++) {           // dreimal blinken19

    bool an = (i % 2 == 0);20

    digitalWrite(LED_BOARD, an);21

    digitalWrite(LED_EXTERN, an);22
    delay(250);23

  }24

  runde++;25

  Serial.printf("\nRunde %d | %s, %d Kerne, %lu MHz, %lu MB Flash\n",26
                runde, ESP.getChipModel(), ESP.getChipCores(),27

                ESP.getCpuFreqMHz(), ESP.getFlashChipSize() / 1048576);28

  int n = WiFi.scanNetworks();            // WLAN-Netze suchen29

  Serial.printf("%d WLAN-Netze gefunden:\n", n);30

  for (int i = 0; i < n && i < 6; i++) {31
    Serial.printf("%2d  %4d dBm  Kanal %2d  %s\n", i + 1,32

                  WiFi.RSSI(i), WiFi.channel(i), WiFi.SSID(i).c_str());33

  }34

  WiFi.scanDelete();35
  delay(10000);36

}37

MAKEROO · support@makeroo.de ESP32 DevKit · MKR-ESP32-005

06 Erweitern: LED am Steckbrett
Der Sketch schaltet GPIO23 schon mit. Mit einer LED und einem Widerstand siehst du, wie du

eigene Bauteile an den ESP32 anschließt.

Verbindungen

ESP32 BAUTEIL

D23 →
220-Ω-Widerstand → LED-Anode

(langes Bein)

GND → LED-Kathode (kurzes Bein)

i Warum 220 Ω?

Bei 3,3 V und einer roten LED fließen damit

rund 6 mA – hell genug und weit unter der

Belastungsgrenze eines GPIO.

! LED bleibt dunkel?

Langes Bein (Anode) zeigt zum Widerstand,

kurzes Bein (Kathode) zu GND. Prüfe, ob das

GND-Kabel wirklich auf der Reihe der

Kathode steckt.

Verdrahtung erstellt mit Fritzing

i Breites Board

Auf einem Steckbrett bleibt neben dem

ESP32 nur eine Lochreihe je Seite frei.

Für größere Aufbauten zwei

Steckbretter nebeneinander nutzen –

oder das Board wie hier daneben legen.

STECKPLATINEN-ANSICHT

SCHALTPLAN

MAKEROO · support@makeroo.de ESP32 DevKit · MKR-ESP32-005

07 Grundlagen in Beispielen
Acht kurze Sketches zeigen, was im ESP32 steckt. Alle liegen auch als Dateien im Download

– öffnen, hochladen, ausprobieren. Den Seriellen Monitor stellst du immer auf 115200 Baud.

Beispiel 1 Taster lesen KEIN ZUBEHÖR

Die Taste BOOT hängt an GPIO0 und lässt sich nach dem Start wie ein normaler Taster abfragen. Jeder Druck

schaltet die blaue LED um.

Kompiliert für „ESP32 Dev Module“: 265 kB Flash (20 %), 22 kB RAM (6 %).

BEISPIELAUSGABE · 115200 BAUD

LED an
LED aus

LED an

Gut zu wissen

INPUT_PULLUP schaltet den internen Pull-up-

Widerstand ein: Offen liest der Pin HIGH,

gedrückt LOW. Einen externen Taster

schließt du genauso zwischen einen GPIO

(z. B. D27) und GND an.

Beispiel 2 LED dimmen mit PWM KEIN ZUBEHÖR

PWM schaltet die LED tausende Male pro Sekunde ein und aus – je länger die Ein-Phase, desto heller. Der ESP32

kann das an fast jedem Ausgangspin.

Kompiliert für „ESP32 Dev Module“: 257 kB Flash (20 %), 22 kB RAM (6 %).

Ergebnis

Die blaue LED „atmet“: rund eine Sekunde

heller, eine Sekunde dunkler.

Gut zu wissen

ledcAttach(pin, frequenz, bits) ist die

Schreibweise ab Board-Paket 3.x.

analogWrite(pin, 0…255) funktioniert

ebenfalls. Mit 50 Hz und 16 Bit steuerst du

so auch Servos.

ESP32_01_Taster.ino

// Beispiel 1: Die BOOT-Taste schaltet die blaue LED um1
const int TASTER = 0;    // BOOT-Taste liegt an GPIO02
const int LED = 2;       // blaue LED "IO2"3
bool ledAn = false;4
bool vorher = HIGH;5
 6
void setup() {7
  Serial.begin(115200);8
  pinMode(TASTER, INPUT_PULLUP);   // offen = HIGH, gedrückt = LOW9
  pinMode(LED, OUTPUT);10
}11
 12
void loop() {13
  bool jetzt = digitalRead(TASTER);14
  if (vorher == HIGH && jetzt == LOW) {   // gerade gedrückt15
    ledAn = !ledAn;16
    digitalWrite(LED, ledAn);17
    Serial.println(ledAn ? "LED an" : "LED aus");18
    delay(50);                            // Taster entprellen19
  }20
  vorher = jetzt;21
}22

ESP32_02_Dimmen.ino

// Beispiel 2: LED weich auf- und abdimmen (PWM)1
const int LED = 2;2
 3
void setup() {4
  ledcAttach(LED, 5000, 8);   // 5 kHz, 8 Bit = Helligkeit 0 bis 2555
}6
 7
void loop() {8
  for (int h = 0; h <= 255; h++) {   // heller werden9
    ledcWrite(LED, h);10
    delay(4);11
  }12
  for (int h = 255; h >= 0; h--) {   // dunkler werden13
    ledcWrite(LED, h);14
    delay(4);15
  }16
}17

MAKEROO · support@makeroo.de ESP32 DevKit · MKR-ESP32-005



· GETTING STARTED · ESP32 DEVKIT SEITE 9

Beispiel 3 Spannung messen POTENTIOMETER 10 KΩ

Der ESP32 misst Spannungen von 0 bis 3,3 V mit 12 Bit. analogReadMilliVolts() rechnet den Rohwert gleich mit den

Kalibrierdaten des Chips in Millivolt um.

Kompiliert für „ESP32 Dev Module“: 271 kB Flash (21 %), 22 kB RAM (6 %).

Aufbau

3V3 → Poti, äußeres Bein

GND → Poti, anderes äußeres Bein

D34 → Poti, Mittelbein (Schleifer)

BEISPIELAUSGABE · 115200 BAUD

Rohwert    0  ->   142 mV
Rohwert 2051  ->  1712 mV

Rohwert 4095  ->  3138 mV

Gut zu wissen

Nimm für Analogwerte die Pins D32–D39

(ADC1). Die ADC2-Pins sind blockiert,

sobald WLAN läuft. Mehr als 3,3 V nur über

einen Spannungsteiler messen.

Beispiel 4 Berührung erkennen DRAHT ODER ALUFOLIE

Zehn Pins des ESP32 sind kapazitive Touch-Sensoren. Ein Stück Draht oder Alufolie an D4 genügt – berührst du es,

sinkt der Messwert deutlich.

Kompiliert für „ESP32 Dev Module“: 278 kB Flash (21 %), 22 kB RAM (6 %).

Aufbau

D4 → Draht, Alufolie oder Münze

BEISPIELAUSGABE · 115200 BAUD

Ruhewert: 74
 73

 72
 15 <- beruehrt
 14 <- beruehrt

Gut zu wissen

Die Schwelle liegt bei 70 % des

Ruhewerts, den der Sketch beim Start

misst. Beim Einschalten das Touch-Feld

also nicht berühren. Touch-fähig: D4, D2,

D15, D13, D12, D14, D27, D33, D32.

ESP32_03_Analog.ino

// Beispiel 3: Spannung messen – Potentiometer an D34 (ADC1)1
const int POTI = 34;2
 3
void setup() {4
  Serial.begin(115200);5
  analogReadResolution(12);              // 12 Bit: 0 bis 40956
}7
 8
void loop() {9
  int roh = analogRead(POTI);10
  int mv = analogReadMilliVolts(POTI);   // kalibriert, in Millivolt11
  Serial.printf("Rohwert %4d  ->  %4d mV\n", roh, mv);12
  delay(500);13
}14

ESP32_04_Touch.ino

// Beispiel 4: Berührung erkennen – Draht an D4 (Touch0)1
const int TOUCH = 4;2
const int LED = 2;3
unsigned long ruhe;4
 5
void setup() {6
  Serial.begin(115200);7
  pinMode(LED, OUTPUT);8
  delay(200);9
  ruhe = touchRead(TOUCH);               // Wert ohne Berührung merken10
  Serial.printf("Ruhewert: %lu\n", ruhe);11
}12
 13
void loop() {14
  unsigned long wert = touchRead(TOUCH);15
  bool beruehrt = wert < ruhe * 7 / 10;  // Finger dran?16
  digitalWrite(LED, beruehrt);17
  Serial.printf("%3lu %s\n", wert, beruehrt ? "<- beruehrt" : "");18
  delay(200);19
}20

MAKEROO · support@makeroo.de ESP32 DevKit · MKR-ESP32-005

08 WLAN & Internet
Beispiel 5 Webseite zum Schalten SMARTPHONE ODER LAPTOP

Der ESP32 öffnet sein eigenes WLAN „MAKEROO-ESP32“ und liefert eine kleine Webseite aus. Damit schaltest du die

blaue LED vom Handy – ganz ohne Router und Internet.

Kompiliert für „ESP32 Dev Module“: 902 kB Flash (70 %), 46 kB RAM (14 %).

192.168.4.1

Die Seite aus dem Sketch im Browser

So testest du

1. Sketch hochladen.

2. Am Handy das WLAN MAKEROO-

ESP32 wählen, Passwort makeroo123.

3. Im Browser 192.168.4.1 öffnen.

4. Auf An und Aus tippen – die blaue

LED folgt.

BEISPIELAUSGABE · 115200 BAUD

Hotspot aktiv: http://192.168.4.1

Gut zu wissen

Das Handy meldet evtl. „Kein Internet“

– einfach verbunden bleiben. Ändere

das Passwort, bevor du das Board

dauerhaft betreibst.

ESP32_05_Webserver.ino

// Beispiel 5: Eigener Hotspot mit Webseite – LED per Handy schalten1
#include <WiFi.h>2
#include <WebServer.h>3
 4
const char* NAME = "MAKEROO-ESP32";    // so heißt das WLAN5
const char* PASSWORT = "makeroo123";   // mindestens 8 Zeichen6
const int LED = 2;7
bool ledAn = false;8
WebServer server(80);9
 10
void seite() {11
  String html = "<!doctype html><meta name='viewport' "12
                "content='width=device-width'><style>body{font-family:"13
                "sans-serif;text-align:center}button{font-size:2em;"14
                "width:40%}</style><h1>MAKEROO ESP32</h1>";15
  html += ledAn ? "<p>LED ist AN</p>" : "<p>LED ist AUS</p>";16
  html += "<a href='/an'><button>An</button></a> "17
          "<a href='/aus'><button>Aus</button></a>";18
  server.send(200, "text/html", html);19
}20
 21
void schalten(bool an) {22
  ledAn = an;23
  digitalWrite(LED, an);24
  seite();25
}26
 27
void setup() {28
  Serial.begin(115200);29
  pinMode(LED, OUTPUT);30
  WiFi.softAP(NAME, PASSWORT);           // ESP32 wird zum Hotspot31
  Serial.print("Hotspot aktiv: http://");32
  Serial.println(WiFi.softAPIP());       // meist 192.168.4.133
  server.on("/", seite);34
  server.on("/an", [] { schalten(true); });35
  server.on("/aus", [] { schalten(false); });36
  server.begin();37
}38
 39
void loop() {40
  server.handleClient();                 // Anfragen beantworten41
}42

MAKEROO · support@makeroo.de ESP32 DevKit · MKR-ESP32-005

Beispiel 6 Heim-WLAN und Uhrzeit DEIN WLAN (2,4 GHZ)

So verbindet sich der ESP32 mit deinem Router und holt die genaue Uhrzeit aus dem Internet – inklusive

automatischer Sommerzeit. Trage in Zeile 4 und 5 deine WLAN-Daten ein.

Kompiliert für „ESP32 Dev Module“: 890 kB Flash (69 %), 47 kB RAM (14 %).

BEISPIELAUSGABE · 115200 BAUD

Verbinde.....

Verbunden, IP-Adresse: 192.168.178.42
20.10.2026  14:52:07
20.10.2026  14:52:08

Gut zu wissen

Die IP-Adresse vergibt dein Router. Mit ihr

erreichst du später einen Webserver im

Heimnetz – kombiniere dafür Beispiel 5 und

6: WiFi.begin() statt WiFi.softAP().

Beispiel 7 Strom sparen mit Deep Sleep KEIN ZUBEHÖR

Im Tiefschlaf ist fast der ganze Chip aus. Ein Timer weckt ihn nach zehn Sekunden, dann beginnt der Sketch wieder

bei setup(). Variablen mit RTC_DATA_ATTR bleiben erhalten.

Kompiliert für „ESP32 Dev Module“: 269 kB Flash (21 %), 22 kB RAM (6 %).

BEISPIELAUSGABE · 115200 BAUD

Aufgewacht! Start Nr. 1

Schlafe 10 Sekunden.
…

Aufgewacht! Start Nr. 2
Schlafe 10 Sekunden.

Gut zu wissen

Zwischen den Meldungen gibt der Chip

selbst ein paar Startzeilen aus (z. B. rst:0x5

(DEEPSLEEP_RESET)) – das ist normal. Auf

dem DevKit verbrauchen USB-Wandler und

LED weiter Strom; für Akkuprojekte zählt vor

allem die lange Schlafzeit.

ESP32_06_WLAN_Uhrzeit.ino

// Beispiel 6: Heim-WLAN und Uhrzeit aus dem Internet1
#include <WiFi.h>2
 3
const char* WLAN_NAME = "DEIN-WLAN";          // anpassen4
const char* WLAN_PASSWORT = "DEIN-PASSWORT";  // anpassen5
 6
void setup() {7
  Serial.begin(115200);8
  WiFi.begin(WLAN_NAME, WLAN_PASSWORT);9
  Serial.print("Verbinde");10
  while (WiFi.status() != WL_CONNECTED) {11
    delay(500);12
    Serial.print(".");13
  }14
  Serial.print("\nVerbunden, IP-Adresse: ");15
  Serial.println(WiFi.localIP());16
  // Zeitzone Deutschland inkl. Sommerzeit, Zeitserver pool.ntp.org17
  configTzTime("CET-1CEST,M3.5.0,M10.5.0/3", "pool.ntp.org");18
}19
 20
void loop() {21
  struct tm zeit;22
  if (getLocalTime(&zeit)) {23
    Serial.println(&zeit, "%d.%m.%Y  %H:%M:%S");24
  }25
  delay(1000);26
}27

ESP32_07_DeepSleep.ino

// Beispiel 7: Deep Sleep – aufwachen, messen, schlafen1
RTC_DATA_ATTR int starts = 0;   // überlebt den Tiefschlaf2
 3
void setup() {4
  Serial.begin(115200);5
  delay(500);6
  starts++;7
  Serial.printf("Aufgewacht! Start Nr. %d\n", starts);8
  // ... hier messen und Werte senden ...9
  Serial.println("Schlafe 10 Sekunden.");10
  Serial.flush();11
  esp_sleep_enable_timer_wakeup(10ULL * 1000000);  // 10 s in µs12
  esp_deep_sleep_start();13
}14
 15
void loop() {16
  // nie erreicht: nach dem Aufwachen startet setup() neu17
}18

MAKEROO · support@makeroo.de ESP32 DevKit · MKR-ESP32-005

09 Sensoren am I2C-Bus
Die meisten Sensoren und Displays sprechen I2C – zwei Leitungen für beliebig viele Geräte.

Beim ESP32 liegt der Bus standardmäßig auf D21 (SDA) und D22 (SCL). Hier am Beispiel

unseres BMP280, der ohnehin 3,3 V braucht.

Beispiel 8 I2C-Scanner I2C-SENSOR, Z. B. BMP280

Der Scanner fragt alle 127 möglichen Adressen ab und listet, wer antwortet. So prüfst du in Sekunden, ob ein Sensor

richtig verkabelt ist und welche Adresse er hat.

Kompiliert für „ESP32 Dev Module“: 287 kB Flash (22 %), 23 kB RAM (7 %).

BEISPIELAUSGABE · 115200 BAUD

Suche I2C-Geraete ...
  gefunden: 0x76

1 Geraet(e) gefunden

STECKPLATINEN-ANSICHT · ESP32 + BMP280

ESP32_08_I2C_Scanner.ino

// Beispiel 8: I2C-Scanner – welche Sensoren hängen am Bus?1
#include <Wire.h>2
 3
void setup() {4
  Serial.begin(115200);5
  Wire.begin(21, 22);               // SDA = D21, SCL = D226
}7
 8
void loop() {9
  int anzahl = 0;10
  Serial.println("Suche I2C-Geraete ...");11
  for (byte adr = 1; adr < 127; adr++) {12
    Wire.beginTransmission(adr);13
    if (Wire.endTransmission() == 0) {   // Geraet antwortet14
      Serial.printf("  gefunden: 0x%02X\n", adr);15
      anzahl++;16
    }17
  }18
  Serial.printf("%d Geraet(e) gefunden\n\n", anzahl);19
  delay(5000);20
}21

MAKEROO · support@makeroo.de ESP32 DevKit · MKR-ESP32-005

10 UNO-Anleitungen auf den ESP32 übertragen
Unsere Sensor-Guides nutzen den UNO R3. Mit dieser Tabelle baust du dieselben

Schaltungen am ESP32 auf – die Sketches laufen meist unverändert, nur die Pinnummern

ändern sich.

AM UNO R3 AM ESP32 WORAUF DU ACHTEN MUSST

5V 3V3
Nur wenn der Sensor 3,3 V verträgt (BMP280, BH1750, VL53L0X,

MPU-6050 …). Braucht er 5 V, an VIN und seinen Ausgang per

Spannungsteiler anpassen.

GND GND Immer verbinden.

A4 (SDA) D21 I2C-Daten; Wire.begin() nutzt D21 automatisch.

A5 (SCL) D22 I2C-Takt.

A0 … A3 D32 … D35
ADC1-Pins, funktionieren auch mit WLAN. Wertebereich 0–4095

statt 0–1023.

D2, D3 D4, D27 Freie digitale Pins ohne Sonderfunktion beim Booten.

D9, D10 D16, D17 Beliebige GPIOs für Trigger/Echo, Daten usw.

SoftwareSerial Serial2
Echte zweite Schnittstelle: Serial2.begin(9600, SERIAL_8N1,

16, 17) – RX an D16, TX an D17.

9600 Baud 115200 Baud
Optional – der ESP32 schafft deutlich mehr. Im Monitor dieselbe

Rate einstellen.

! Kein 5-V-Signal an die GPIOs

Sensoren mit 5-V-Ausgang (z. B. HC-SR04-Echo, HC-SR501) über einen Spannungsteiler

anschließen, etwa 1 kΩ in Reihe und 2 kΩ nach GND. Das senkt 5 V auf ungefähr 3,3 V.

i Bibliotheken

Adafruit-, Pololu- und die meisten anderen Arduino-Bibliotheken unterstützen den ESP32. Du

installierst sie wie gewohnt über den Bibliotheksverwalter.

MAKEROO · support@makeroo.de ESP32 DevKit · MKR-ESP32-005

11 Wenn es nicht klappt
SYMPTOM LÖSUNG

Kein Port in der IDE Anderes USB-C-Kabel probieren – viele Kabel laden nur. Danach den

CH340-Treiber installieren (Schritt 2).

„Failed to connect to ESP32“ bzw.

„Timed out … packet header“

Beim Hochladen die Taste BOOT gedrückt halten, bis „Connecting…“ in

Prozente übergeht. Hilft das nicht: Werkzeuge → Upload Speed → 115200.

Nur wirre Zeichen im Monitor Baudrate unten rechts auf 115200 stellen.

Keine Startmeldung Der Monitor war beim Start noch zu. Taste EN drücken – der ESP32 startet

neu.

„Brownout detector was triggered“,

ständige Neustarts

Der USB-Port liefert zu wenig Strom. Anderen Port, aktiven Hub oder

kürzeres Kabel nutzen.

0 WLAN-Netze gefunden Der ESP32 sieht nur 2,4-GHz-Netze. Board nicht auf Metall legen, Antenne

(Leiterbahn am Modulende) frei lassen.

Upload klappt nur ohne Schaltung An D12, D2, D15 oder D5 hängt etwas, das den Boot-Pegel verändert. Diese

Pins beim Hochladen frei lassen.

12 Wie geht's weiter?
Mit ESPHome & Home Assistant

Ohne eigenen Code: So erscheint die blaue LED als

Schalter in Home Assistant.

esphome:

  name: makeroo-esp32

esp32:

  board: esp32dev

wifi:

  ssid: !secret wifi_ssid

  password: !secret wifi_password

api:

output:

  - {platform: gpio, pin: GPIO2, id: led}

light:

  - {platform: binary, name: LED, output: led}

Projektideen

WLAN-Thermometer

DHT22 an D4, Werte im Browser oder in Home Assistant

Handy-Schalter

Relais an D23, per Webseite vom Smartphone schalten

Messknoten mit Akku

Messen, senden, Deep Sleep – monatelang autark

Passt dazu

ESP32 Gehäuse aus

Acryl

Schützt das Board

MKR-ESP32-HOUSING-045

0,96" OLED-Display

Werte direkt anzeigen

MKR-0.96-OLED-015

DHT22 Temperatur &

Feuchte

Erster WLAN-Sensor

MKR-DHT22-009

Relais KY-019

Verbraucher per WLAN

schalten

MKR-KY019-034

Sicherheitshinweise. Elektronikmodul für Ausbildungs-,

Entwicklungs- und Experimentierzwecke. Nur mit Kleinspannung

betreiben, niemals an Netzspannung. Nicht geeignet für Kinder unter

14 Jahren; Nutzung nur unter Aufsicht. Elektroaltgeräte gehören nicht

in den Hausmüll – Rückgabe kostenlos bei kommunalen

Sammelstellen.

Verantwortlicher Wirtschaftsakteur (GPSR): Alexander Prez ·

makeroo · Hinterm Bahnhof 4 · 90513 Zirndorf · support@makeroo.de

Fragen? Schreib uns mit einem Foto deines Aufbaus.

Arduino ist eine Marke der Arduino SA, ESP32 und ESPHome sind

Marken ihrer Inhaber und dienen nur der Kompatibilitätsangabe.

